

Первое знакомство с радикалами

С точки зрения Maple радикалы представляют собой фундаментальный тип данных и все операции над ними выполняются точно.

Если, например, нас интересует квадратный корень из 2, то следует написать

```
> sqrt(2);
```

$$\sqrt{2}$$

Как мы видим, никакого преобразования в десятичную дробь не происходит.

Уже на этапе ввода, система Maple способна выполнять автоупрощения.

```
> sqrt(8);
```

$$2\sqrt{2}$$

В подавляющем большинстве случаев это следует приветствовать. Однако если пользователь хочет сохранить выражение в исходном виде, то в некоторых случаях подавление автоупрощений может потребовать таких действий, которые неочевидны для начинающего.

Простейший способ временного "замораживания" выражений состоит в использовании одинарных кавычек.

```
> 'sqrt(8)';
```

$$\sqrt{8}$$

Из сказанного выше следует, что в том случае, когда под знаком корня оказывается величина, являющаяся полным квадратом, то в результате автоупрощений результат не будет содержать радикалов.

```
> sqrt(9)+sqrt(4);
```

$$5$$

Для радикалов произвольных степеней следует использовать символ возведения в степень, который, обычно, имеет вид "^".

Упомянутый выше квадратный корень из 2 можно записать и так

```
> 2^(1/2);
```

$$\sqrt{2}$$

Не следует забывать о *приоритете* операций, поскольку попытка сэкономить на скобках приведёт к неожиданному результату.

```
> 2^1/2;
```

$$1$$

Если символ "птичка" вам по каким-либо причинам не нравится, то, как и в языке *Fortran*, для возведения в степень можно использовать пару записанных подряд знаков умножения.

```
> 2**(1/2);
```

$$\sqrt{2}$$

Специальной функции для извлечения кубических корней, подобной *sqrt*, в системе Maple нет. Для извлечения кубического корня из 2 приходится писать так

```
> 2^(1/3);
```

$$2^{1/3}$$

Следует помнить, что радикалы являются многозначными функциями, а значок "^" приводит к вычислению главного значения корня. Такая трактовка может показаться неожиданной для тех пользователей, которые привыкли к тому, что извлечение корня нечётной степени из отрицательного числа даёт в результате отрицательное вещественное число.

Хрестоматийный пример состоит в вычислении кубического корня из -1. Пусть

```
> v:=(-1)^(1/3);
```

$$v := (-1)^{1/3}$$

Представим эту величину в виде суммы вещественной и мнимой частей

```
> evalc(v);
```

$$\frac{1}{2} + \frac{I\sqrt{3}}{2}$$

Нетрудно заметить, что данная величина равна $e^{\frac{1}{3}i\pi}$

```
> exp(I*Pi/3);
```

$$\frac{1}{2} + \frac{I\sqrt{3}}{2}$$

Если необходимо выполнить серию преобразований по правилам элементарной математики, то имеет смысл записывать радикалы с помощью функции *surd*. Первым аргументом этой функции является выражение, из которого мы хотим извлечь корень. Степень самого корня указывается вторым аргументом.

```
> surd(-1,3);surd(1-9,3);
```

-1
-2

Для решения задач по элементарной математике, то есть в тех случаях, когда вычисления выполняются в поле вещественных чисел, можно использовать возможности пакета *RealDomain*. После его загрузки для ограниченного набора функций Maple создаётся окружение, в котором аргументы всех этих функций считаются вещественными. Конечный результат также преобразуется к вещественному виду и если такое преобразование невозможно, то полагают, что результат неопределён.

Проиллюстрируем особенности применения *RealDomain* на простом примере.

```
> sqrt(-2);
```

$I\sqrt{2}$

Если теперь загрузить этот пакет

```
> with(RealDomain);
```

```
[ℑ, ℔, ^, arccos, arccosh, arccot, arccoth, arccsc, arccsch, arcsec, arcsech, arcsin, arcsinh, arctan, arctanh, cos, cosh, cot, coth, csc, csch, eval, exp, expand, limit, ln, log, sec, sech, signum, simplify, sin, sinh, solve, sqrt, surd, tan, tanh]
```

то в списке загруженных функций можно увидеть *sqrt*.

Попробуем ещё раз вычислить квадратный корень из -2.

```
> sqrt(-2);
```

undefined

Операция возведения в степень тоже оказалась в списке загруженных функций. Это даёт надежду на то, что корни нечётных степеней будут извлекаться так же, как учили в школе.

```
> (-1)^(1/3);(-1)^(1/5);
```

-1
-1

Если расчёты с радикалами по правилам элементарной математики выполняются лишь эпизодически, то загрузка *RealDomain* на постоянной основе нецелесообразна. Происходит так называемое "затенение" нескольких десятков функций и, тем самым, затрудняется доступ к их оригинальным версиям.

Оказывается, что окружение *RealDomain* можно создавать и в локальном контексте.

Покажем как это сделать, предварительно очистив память компьютера.

```
> restart;
```

Убедимся, что не существует вещественного значения корня из -2.

```
> use RealDomain in sqrt(-2) end;
```

undefined

За пределами этого окружения функция *sqrt* ведёт себя вполне обычным образом.

```
> sqrt(-2);
```

$I\sqrt{2}$

Внутри окружения **RealDomain** можно разместить сразу несколько операторов.

```
> use RealDomain in v:=(-1)^(1/3);w:=(-32)^(1/5) end;  
v := -1  
w := -2
```

Корни из чисел с плавающей точкой можно находить с помощью операции "[^]" так же, как и при проведении точных вычислений. Результат будет моделировать работу калькулятора.

```
> 3.0^(1/2);  
1.732050808
```

Приближённое значение корня может быть найдено и при помощи функции *surd*. Выбирайте сами, что вам больше нравится.

```
> surd(-3.375,3);  
-1.500000000
```

Помимо двух вышеупомянутых способов записи радикалов в Maple существует функция *root*, которая предназначена для аналогичных целей. Эта функция возвращает главное значение корня численного выражения, которое может быть как точным, так и приближённым.

В наиболее употребительной форме записи *root* имеет такие же аргументы, что и функция *surd*. Первый аргумент представляет собой выражение, из которого мы хотим извлечь корень. Степень самого корня является вторым аргументом.

Неоднократно встречавшийся ранее квадратный корень из 2 записывается следующим образом:

```
> root(2,2);  
 $\sqrt{2}$ 
```

Кубический корень из -1 будет выглядеть так

```
> v:=root(-1,3);  
v := (-1)1/3
```

Поскольку речь идёт о главном значении, то, в отличие от *surd*, результат будет комплексным.

```
> evalc(v);  
 $\frac{1}{2} + \frac{1\sqrt{3}}{2}$ 
```

Некоторые команды Maple допускают альтернативные формы записи. К их числу относится и *root*. Степень корня можно записать в квадратных скобках сразу после имени команды. Тогда предыдущий пример с кубическим корнем из -1 преобразуется к виду

```
> v:=root[3](-1);  
v := (-1)1/3
```

Если подкоренное выражение имеет тип *float*,

```
> v:=1/3.0;  
whattype(v);  
v := 0.3333333333  
float
```

то корень извлекается приближённо.

```
> root(v,2);  
0.5773502692
```

Тонкие различия между результатами, которые выдают *root*, *surd* и "[^]", обусловлены деталями упрощения подкоренных выражений.

Некоторые преобразования Maple выполняет автоматически. Например, при возведении рационального числа *a* в рациональную степень $\frac{n}{d}$, при условии, что $n = q \cdot d + r$, где

$0 < r < d$, будет автоматически выполнено преобразование $a^{\frac{n}{d}} = a^q \cdot a^{\frac{r}{d}}$. В результате возникают следующие нюансы:

```
> 2^(5/2);(2^5)^(1/2);
```

$$4\sqrt{2}$$
$$\sqrt{32}$$

Как мы видим, *root* и *surd* проделывают дополнительную работу, которая, впрочем, может оказаться бесполезной.

```
> root(2^5,2);surd(2^5,2);
```

$$4\sqrt{2}$$
$$4\sqrt{2}$$

Сравним результаты работы этих функций в случае, когда аргументом является отрицательное число.

```
> (-2^7)^(1/3);root[3](-2^7);surd(-2^7,3);
```

$$(-128)^{1/3}$$
$$4(-1)^{1/3}2^{1/3}$$
$$-42^{1/3}$$

Полезные возможности команды *convert*, связанные с радикалами

Неоднозначность записи радикалов требует наличия в системе Maple средств, позволяющих трансформировать радикалы из одного представления в другое в зависимости от потребностей решаемой задачи.

Если радикал задан посредством операции "^" или же путём обращения к функции *root*, то его можно преобразовать в *surd*, используя уже известную команду *convert*.

```
> v:=(-64)^(1/3);
```

$$v := (-64)^{1/3}$$

```
> convert(v,surd);
```

$$-4$$

Отметим, что в результате подобного преобразования величина корня не обязана сохраняться. В данном случае имеем

```
> evalc(v);
```

$$\frac{64^{1/3}}{2} + \frac{164^{1/3}\sqrt{3}}{2}$$

Когда подкоренное выражение может быть представлено в виде степени с

рациональным показателем вида $x^{\frac{p}{q}}$, то команда $\text{convert}\left(x^{\frac{p}{q}}, \text{surd}\right)$ преобразует результат к виду $\text{surd}(x, q)^p$.

```
> convert(x^(2/3),surd);
```

$$\left(\sqrt[3]{x}\right)^2$$

Команда *convert* поддерживает и обратное преобразование. Радикалы, записанные с помощью *surd* можно привести к виду, который получается, если использовать операцию "^". Для выполнения такого преобразования следует указать опцию *power*.

```
> v:=surd(2*x,3);
```

$$v := 2^{1/3}\sqrt[3]{x}$$

```
> convert(v,power);
```

$$2^{1/3}x^{1/3}$$

Как и в предыдущем случае такое преобразование не обязательно сохранять результат, если в подкоренном выражении присутствуют переменные


```
> v1:=surd(x,3);v2:=convert(v1,power);
```

$$v1 := \sqrt[3]{x}$$
$$v2 := x^{1/3}$$

Переменная *v1* определена для всех вещественных значений *x*. На графике её

значения отображаются красным цветом. В свою очередь, v_2 определена только для положительных x , перерисовывая правую ветвь v_1 синим цветом. При $x < 0$ v_2 принимает комплексные значения и на графике не показывается.

```
> plot([v1,v2],x=-2..2,color=[red,blue],size=[200,200]);
```



В радикалах выражаются значения тригонометрических функций стандартных углов. Рассмотрим хорошо известный из школьного курса пример. Пусть угол α равен 60° .

```
> alpha:=convert(60*degrees,radians);
```

$$\alpha := \frac{\pi}{3}$$

Синус этого угла Maple выводит без лишних напоминаний.

```
> sin(alpha);
```

$$\frac{\sqrt{3}}{2}$$

Предположим теперь, что мы хотим узнать, чему равен синус угла 15° . Повторим вычисления.

```
> beta:=convert(15*degrees,radians);
```

$$\beta := \frac{\pi}{12}$$

```
> sin(beta);
```

$$\sin\left(\frac{\pi}{12}\right)$$

Maple отказывается выводить ответ, хотя и знает его. Попробуем "подтолкнуть" систему компьютерной алгебры, применив команду *convert* с опцией *radical*.

```
> convert(sin(beta),radical);
```

$$\frac{\sqrt{2}(\sqrt{3}-1)}{4}$$

Возможности системы не ограничены вычислением только функции синус. Найдём тангенс угла 15° .

```
> convert(tan(beta),radical);
```

$$2 - \sqrt{3}$$

Реализованные в Maple алгоритмы позволяют выразить в радикалах значения тригонометрических функций в тех случаях, когда их аргумент представим в виде $\frac{n}{120} \pi$ для некоторого целого n .

Нетрудно заметить, что в это семейство попадает угол 36° .

```
> alpha:=convert(36*degrees,radians);
```

$$\alpha := \frac{\pi}{5}$$